* define interaction terms -----------------------------------------------------

gen single_pt = treatment * akp_jun
gen multi_pt = treatment_multi * akp_jun

* label the terms --------------------------------------------------------------

label variable single_pt           "Treatment X Pre-test"
label variable multi_pt            "Multiple Treatment X Pre-test"

* run the regressions models and save them to memory ---------------------------

eststo: reg akp_nov treatment single_pt akp_jun recruitment_pool,               ///
cluster(province)

eststo: reg akp_nov treatment single_pt akp_jun recruitment_pool                ///
treatment_nonterror casualty_district kurdish_district akp_district             ///
higher_education min_margin turnout_jun, cluster(province)

eststo: reg akp_nov treatment treatment_multi multi_pt akp_jun                  ///
recruitment_pool, cluster(province)

eststo: reg akp_nov treatment treatment_multi multi_pt akp_jun recruitment_pool ///
treatment_nonterror casualty_district kurdish_district akp_district             ///
higher_education min_margin turnout_jun, cluster(province)

* print the results ------------------------------------------------------------

esttab,                                                                         ///
replace se b(3) se(3) stats(N vce r2) label nodepvars nonotes nomtitles         ///
order(treatment treatment_multi single_pt multi_pt)                             ///
substitute(vce "Clusters" cluster "81")                                         ///
title("Table S11: Regression models with interaction terms, constructed with Pre-test")

* save the results -------------------------------------------------------------

esttab using "./tables/table_s11.tex",                                          ///
replace se b(3) se(3) stats(N vce r2) label nodepvars nonotes nomtitles         ///
order(treatment treatment_multi single_pt multi_pt)                             ///
substitute(vce "Clusters" cluster "81")                                         ///
title("Regression models with interaction terms, constructed with \textit{Pre-test}")

* remove the interaction variables ---------------------------------------------

drop single_pt multi_pt

* remove the models from memory ------------------------------------------------

eststo clear